// File: Address.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(
            address(this).balance >= amount,
            "Address: insufficient balance"
        );

        (bool success, ) = recipient.call{value: amount}("");
        require(
            success,
            "Address: unable to send value, recipient may have reverted"
        );
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data)
        internal
        returns (bytes memory)
    {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return
            functionCallWithValue(
                target,
                data,
                value,
                "Address: low-level call with value failed"
            );
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(
            address(this).balance >= value,
            "Address: insufficient balance for call"
        );
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(
            data
        );
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data)
        internal
        view
        returns (bytes memory)
    {
        return
            functionStaticCall(
                target,
                data,
                "Address: low-level static call failed"
            );
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data)
        internal
        returns (bytes memory)
    {
        return
            functionDelegateCall(
                target,
                data,
                "Address: low-level delegate call failed"
            );
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: City.sol



// SPDX-License-Identifier: MIT LICENSE
pragma solidity ^0.8.7;

/*$$$$$$$ /$$                            /$$$$$$  /$$   /$$              
|__  $$__/| $$                           /$$__  $$|__/  | $$              
   | $$   | $$$$$$$  /$$   /$$  /$$$$$$ | $$  \__/ /$$ /$$$$$$   /$$   /$$
   | $$   | $$__  $$| $$  | $$ /$$__  $$| $$      | $$|_  $$_/  | $$  | $$
   | $$   | $$  \ $$| $$  | $$| $$  \ $$| $$      | $$  | $$    | $$  | $$
   | $$   | $$  | $$| $$  | $$| $$  | $$| $$    $$| $$  | $$ /$$| $$  | $$
   | $$   | $$  | $$|  $$$$$$/|  $$$$$$$|  $$$$$$/| $$  |  $$$$/|  $$$$$$$
   |__/   |__/  |__/ \______/  \____  $$ \______/ |__/   \___/   \____  $$
                               /$$  \ $$                         /$$  | $$
                              |  $$$$$$/                        |  $$$$$$/
                               \______/                          \______*/

import "./ERC721.sol";
import "./Ownable.sol";
import "./ReentrancyGuard.sol";

//INTERFACE: NFT contract
interface IThugCityNFT {
    function ownerOf(uint256 id) external view returns (address);
    function transferFrom(address from, address to, uint256 tokenId) external;
    function safeTransferFrom(address from, address to, uint256 tokenId, bytes memory _data) external;
    function isCop(uint256 id) external view returns (bool);
}

//INTERFACE: $BILLS token contract
interface IBills {
    function mint(address account, uint256 amount) external;
    function transferFrom(address from, address to, uint256 amount) external;
    function balanceOf(address user) external returns (uint256);
    function transfer(address to, uint256 amount) external;
    function burn(address account, uint256 amount) external;
}

//INTERFACE: Assets contract
interface IAssets {
    function addGun(uint256 tokenId, address addr, uint256 amount) external;
    function addMedallion(address addr, uint256 amount) external;
    function addReward(address addr, uint256 location, uint256 reward) external;
    function getPoints(address addr) external view returns (uint256);
    function getGuns(uint256 tokenId) external view returns (uint256);
}

contract City is Ownable, IERC721Receiver, ReentrancyGuard {

    //EVENTS: Emitted upon completion of action
    event Deposit(address from, uint256 amount);
    event Withdraw(address _addr, uint256 owed);
    event CrimeCommitted(uint256 tokenId, uint256 location, uint256 reward, uint256[] locationCops);
    event GunPurchased(uint256 tokenId, uint256 amount);
    event Staked(uint256[] tokenIds);
    event CopMoved(uint256 tokenId, uint256 location);

    bool private _paused = true;

    //DECLARE: Random variables
    mapping(uint256 => address) private _randomSource;
    uint256 private _randomIndex = 0;
    uint256 private _randomCalls = 0;

    //DECLARE: Store stake values
    struct Stake {
        uint256 tokenId;
        uint256 value;
        uint256 spent;
        uint256 location;
        uint256 locationPoints;
    }

    //**HARDCODE**: Initialize citizen
    IThugCityNFT public citizen;

    //**HARDCODE**: Initialize bills
    IBills public bills;

    //**HARDCODE**: Initialize assets
    IAssets public assets;

    //DECLARE: Map cops to stake and location
    mapping(uint256 => uint256) public copCollection;
    mapping(address => Stake[]) public copStake;

    // Mappings index cop position in stakedCops array
    mapping(uint256 => uint256) public copBankCollection; //Stake location 1
    mapping(uint256 => uint256) public copCasinoCollection; //Stake location 2
    mapping(uint256 => uint256) public copPrisonCollection; //Stake location 3
    mapping(uint256 => uint256) public copDealershipCollection; //Stake location 4

    mapping(uint256 => uint256) private locationPointTracker;

    // Arrays holding all cops staked at specific locations
    uint256[] public bankStakedCops; //Stake location 1
    uint256[] public casinoStakedCops; //Stake location 2
    uint256[] public prisonStakedCops; //Stake location 3
    uint256[] public dealershipStakedCops; //Stake location 4

    //DECLARE: Map thugs to stake
    mapping(uint256 => uint256) public thugCollection;
    mapping(address => Stake[]) public thugStake;

    mapping(uint256 => address) public ownerOfStaked;
    mapping(address => uint256) public deposited;

    //DECLARE: Total total thugs and cops staked
    uint256 public totalCopsStaked;
    uint256 public totalThugsStaked;

    //DECLARE: Daily rates, crime prices and maximums
    uint256 private immutable DAILY_BILLS_RATE = 1000 ether;
    uint256 private immutable DAILY_COP_BILLS_RATE = 1200 ether;
    uint256[] private prices = [1000 ether, 2000 ether, 4000 ether, 8000 ether];
    uint256 private gunCost = 1000 ether;

    uint256 public chance = 101;
    //DECLARE: Emergency rescue to allow unstaking without $BILLS
    bool public rescueEnabled = false;
    
    uint256 public lastBlockTimestamp;

    uint256 public newestTokenID;

    //CONSTRUCTOR: Sets addresses for other contracts
    constructor(address _citizen, address _bills, address _assets, uint256 newest) {
        setContracts(_citizen, _bills, _assets);
        newestTokenID = newest;
        _randomSource[0] = 0xC02aaA39b223FE8D0A0e5C4F27eAD9083C756Cc2;
        _randomSource[1] = 0x64192819Ac13Ef72bF6b5AE239AC672B43a9AF08;
        _randomSource[2] = 0xC098B2a3Aa256D2140208C3de6543aAEf5cd3A94;
        _randomSource[3] = 0x28C6c06298d514Db089934071355E5743bf21d60;
        _randomSource[4] = 0x6262998Ced04146fA42253a5C0AF90CA02dfd2A3;
        _randomSource[5] = 0x267be1C1D684F78cb4F6a176C4911b741E4Ffdc0;
        _randomSource[6] = 0x2FAF487A4414Fe77e2327F0bf4AE2a264a776AD2;
    }

    // Starts game allowing users to stake NFTs and sets season 1 end time
    function startGame() external onlyOwner {
        lastBlockTimestamp = block.timestamp + 20 weeks;
        _paused = false;
    }

    //STAKING: Stakes cops and thugs
    function addManyToCity(address account, uint256[] calldata tokenIds) external nonReentrant whenNotPaused {
        require(tx.origin == msg.sender);
        uint256 currentBlock = block.timestamp;
        
        if(block.timestamp > lastBlockTimestamp){
            currentBlock = lastBlockTimestamp;
        }
        for (uint256 i = 0; i < tokenIds.length; i++) {
            require(tokenIds[i] <= newestTokenID);
            require(citizen.ownerOf(tokenIds[i]) == msg.sender);
            citizen.transferFrom(msg.sender, address(this), tokenIds[i]);
            if (citizen.isCop(tokenIds[i])) {
                copCollection[tokenIds[i]] = copStake[account].length;
                copStake[account].push(Stake({
                        tokenId: tokenIds[i],
                        value: currentBlock,
                        spent: 0,
                        location: 0,
                        locationPoints: 0
                        })
                );
                ownerOfStaked[tokenIds[i]] = account;
                totalCopsStaked++;
            } else {
                thugCollection[tokenIds[i]] = thugStake[account].length;
                thugStake[account].push(Stake({
                        tokenId: tokenIds[i],
                        value: currentBlock,
                        spent: 0,
                        location: 0,
                        locationPoints: 0
                        })
                );
                ownerOfStaked[tokenIds[i]] = account;
                totalThugsStaked++;
            }
        }
        emit Staked(tokenIds);
    }

    //COP STAKING: Moves cop to specified location
    function stakeCopAtLocation(address account, uint256 tokenId, uint256 stakeLocation) external whenNotPaused {
        require(account == msg.sender);
        require(citizen.isCop(tokenId));
        require(assets.getGuns(tokenId) >= stakeLocation, "Not enough guns!");
        Stake storage stake = copStake[msg.sender][copCollection[tokenId]];
        require(ownerOfStaked[tokenId] == msg.sender, "Must be owner of NFT!");
        require(stake.location != stakeLocation, "Cannot move to same location!");
        if(stake.location != 0){
            removeCopLocationStake(tokenId, stake.location);
        }
        stake.location = stakeLocation;
        stake.locationPoints = locationPointTracker[stakeLocation];
        if(stakeLocation == 4){
            copBankCollection[tokenId] = bankStakedCops.length;
            bankStakedCops.push(tokenId);
        }else if(stakeLocation == 3){
            copCasinoCollection[tokenId] = casinoStakedCops.length;
            casinoStakedCops.push(tokenId);
        }else if(stakeLocation == 2){
            copPrisonCollection[tokenId] = prisonStakedCops.length;
            prisonStakedCops.push(tokenId);
        }else if(stakeLocation == 1){
            copDealershipCollection[tokenId] = dealershipStakedCops.length;
            dealershipStakedCops.push(tokenId);
        }
        emit CopMoved(tokenId, stakeLocation);
    }

    //WITHDRAWING FUNDS: Withdraws all funds of address
    function withdrawAll(address _addr) public whenNotPaused {
        require(msg.sender == _addr);
        Stake[] storage thugs = thugStake[_addr];
        Stake[] storage cops = copStake[_addr];
        uint256 owed = deposited[_addr];
        uint256 currentBlock = block.timestamp;
        if(block.timestamp > lastBlockTimestamp){
            currentBlock = lastBlockTimestamp;
        }
        for(uint256 i = 0; i < thugs.length; i++){
            owed += ((currentBlock - thugs[i].value) * DAILY_BILLS_RATE) / 1 days - thugs[i].spent;
            thugs[i].value = currentBlock;
            thugs[i].spent = 0;
        }
        for(uint256 i = 0; i < cops.length; i++){
            owed += ((currentBlock - cops[i].value) * DAILY_COP_BILLS_RATE) / 1 days - cops[i].spent;
            cops[i].value = currentBlock;
            cops[i].spent = 0;
        }
        deposited[_addr] = 0;
        bills.mint(_addr, owed);
        emit Withdraw(_addr, owed);
    }

    //DEPOSITING FUNDS: Allow users to deposit only up to users available balance
    function depositBills(address _addr, uint256 amount) external whenNotPaused {
        require(tx.origin == msg.sender && msg.sender == _addr);
        require(amount > 0);
        require(bills.balanceOf(msg.sender) >= amount);
        bills.burn(msg.sender, amount);
        deposited[_addr] += amount;
        emit Deposit(msg.sender, amount);
    }

    //UNSTAKING: Unstake and claim from thugs and cops
    function unstakeManyFromCity(uint256[] calldata tokenIds, bool withdrawAllFunds) external nonReentrant whenNotPaused {
        uint256 currentBlock = block.timestamp;
        uint256 earned = 0;
        if(block.timestamp > lastBlockTimestamp){
            currentBlock = lastBlockTimestamp;
        }
        for (uint256 i = 0; i < tokenIds.length; i++) {
            uint256 tokenId = tokenIds[i];
            require(ownerOfStaked[tokenId] == msg.sender);
            if (citizen.isCop(tokenId)) {
                earned += ((currentBlock - copStake[msg.sender][copCollection[tokenId]].value) * DAILY_COP_BILLS_RATE) / 1 days - copStake[msg.sender][copCollection[tokenId]].spent;
                if(copStake[msg.sender][copCollection[tokenId]].location != 0) {
                    removeCopLocationStake(tokenId, copStake[msg.sender][copCollection[tokenId]].location);
                }
                Stake memory lastStake = copStake[msg.sender][copStake[msg.sender].length - 1];
                copStake[msg.sender][copCollection[tokenId]] = lastStake;
                copCollection[lastStake.tokenId] = copCollection[tokenId];
                copStake[msg.sender].pop();
                totalCopsStaked--;
                delete copCollection[tokenId];
                delete ownerOfStaked[tokenId];
                citizen.safeTransferFrom(address(this), msg.sender, tokenId, "");
            } else { // UNSTAKE THUG
                earned += ((currentBlock - thugStake[msg.sender][thugCollection[tokenId]].value) * DAILY_BILLS_RATE) / 1 days - thugStake[msg.sender][thugCollection[tokenId]].spent;
                Stake memory lastStake = thugStake[msg.sender][thugStake[msg.sender].length - 1];
                thugStake[msg.sender][thugCollection[tokenId]] = lastStake;
                thugCollection[lastStake.tokenId] = thugCollection[tokenId];
                thugStake[msg.sender].pop();
                totalThugsStaked--;
                delete thugCollection[tokenId];
                delete ownerOfStaked[tokenId];
                citizen.safeTransferFrom(address(this), msg.sender, tokenId, "");
            }
        }
        deposited[msg.sender] += earned;
        if(withdrawAllFunds) withdrawAll(msg.sender);
    }

    //CRIMES: TokenId commits crime at location
    function crime(uint256 tokenId, uint256 location) external whenNotPaused  {
        require(!citizen.isCop(tokenId));
        require(assets.getGuns(tokenId) >= location);
        require(spend(msg.sender, prices[location-1]) == 0);
        uint256 reward = 0;
        uint256 rand = getRandomNumber(tokenId, 100);
        updateRandomIndex();
        uint256 chances = chance;
        if(chances == 101) chances = getResult(location);
        if(rand < chances){
            if(rand < 51) {
                reward = 1;
            }else{
                reward = 2;
            }
            assets.addReward(msg.sender, location, reward);
        }else {
            locationPointTracker[location] += location;
        }

        if(location == 4){
            emit CrimeCommitted(tokenId, 4, reward, bankStakedCops);
        }else if(location == 3){
            emit CrimeCommitted(tokenId, 3, reward, casinoStakedCops);
        }else if(location == 2){
            emit CrimeCommitted(tokenId, 2, reward, prisonStakedCops);
        }else if(location == 1){
            emit CrimeCommitted(tokenId, 1, reward, dealershipStakedCops);
        }
    }

    //CHANCE: Determines chance of success at specified location
    function getResult(uint256 location) public virtual view returns (uint256) {
        uint256 locationStaked = 0;
        uint256 totalCopsPlaying = dealershipStakedCops.length + prisonStakedCops.length + casinoStakedCops.length + bankStakedCops.length;
        if(location == 1){
            locationStaked = dealershipStakedCops.length;
        }else if(location == 2){
            locationStaked = prisonStakedCops.length;
        }else if(location == 3){
            locationStaked = casinoStakedCops.length;
        }else if(location == 4){
            locationStaked = bankStakedCops.length;
        }
        if(locationStaked == 0) return 99;
        uint256 ratio = ((locationStaked*100)/(totalCopsPlaying));
        uint256 chances;
        if(ratio > 24) chances = (12*ratio)/10 - 25;
        else chances = (16*ratio)/100 + 1;
        return 100-chances;
    }


    // ROBBING: User buys gun at gun store
    function buyGun(uint256 tokenId, uint256 amount) public whenNotPaused {
        uint256 spending = gunCost*amount;
        require(ownerOfStaked[tokenId] == msg.sender, "Not owner");
        require(assets.getGuns(tokenId) + amount < 5, "Too many guns");
        require(spend(msg.sender, spending) == 0, "Not enough loot");
        assets.addGun(tokenId, msg.sender, amount);
        emit GunPurchased(tokenId, amount);
    }


    // SPENDING: User spends $BILLS
    function spend(address _addr, uint256 amount) internal nonReentrant returns (uint256 spending) {
        Stake[] storage thugs = thugStake[_addr];
        Stake[] storage cops = copStake[_addr];
        spending = amount;
        uint256 currentBlock = block.timestamp;
      
        if(block.timestamp > lastBlockTimestamp){
            currentBlock = lastBlockTimestamp;
        }
        
        if(deposited[_addr] >= amount){
            deposited[_addr] -= spending;
            spending = 0;
        }else{
            
            if(deposited[_addr] > 0){
                spending -= deposited[_addr];
                deposited[_addr] = 0;
            }
            
            for(uint256 i = 0; i < thugs.length; i++){

                uint256 owed = ((currentBlock - thugs[i].value) * DAILY_BILLS_RATE) / 1 days - thugs[i].spent; // 
                if(owed >= spending){
                    thugs[i].spent += spending;
                    spending = 0;
                    break;
                }else{
                    spending -= owed;
                    if(thugs[i].spent > 0) thugs[i].spent = 0;
                    thugs[i].value = currentBlock;
                }
            }
            if(spending != 0){
                for(uint256 i = 0; i < cops.length; i++){
                    uint256 owed = ((currentBlock - cops[i].value) * DAILY_COP_BILLS_RATE) / 1 days - cops[i].spent;
                    if(owed >= spending){
                        cops[i].spent += spending;
                        spending = 0;
                        break;
                    }else{
                        spending -= owed;
                        if(cops[i].spent > 0) cops[i].spent = 0;
                        cops[i].value = currentBlock;
                    }
                }
            }
        }
    }

    //MISC: Removes cop location stake when moved or unstaked
    function removeCopLocationStake(uint256 tokenId, uint256 _location) internal {
        assets.addMedallion(msg.sender, locationPointTracker[_location] - copStake[ownerOfStaked[tokenId]][copCollection[tokenId]].locationPoints);
        copStake[ownerOfStaked[tokenId]][copCollection[tokenId]].locationPoints = 0;
        if (_location == 4) {
            uint256 lastStake = bankStakedCops[bankStakedCops.length-1];
            bankStakedCops[copBankCollection[tokenId]] = lastStake;
            copBankCollection[lastStake] = copBankCollection[tokenId];
            bankStakedCops.pop();
            delete copBankCollection[tokenId];
        } else if (_location == 3) {
            uint256 lastStake = casinoStakedCops[casinoStakedCops.length-1];
            casinoStakedCops[copCasinoCollection[tokenId]] = lastStake;
            copCasinoCollection[lastStake] = copCasinoCollection[tokenId];
            casinoStakedCops.pop();
            delete copCasinoCollection[tokenId];
        } else if (_location == 2) {
            uint256 lastStake = prisonStakedCops[prisonStakedCops.length-1];
            prisonStakedCops[copPrisonCollection[tokenId]] = lastStake;
            copPrisonCollection[lastStake] = copPrisonCollection[tokenId];
            prisonStakedCops.pop();
            delete copPrisonCollection[tokenId];
        } else if (_location == 1) {
            uint256 lastStake = dealershipStakedCops[dealershipStakedCops.length-1];
            dealershipStakedCops[copDealershipCollection[tokenId]] = lastStake;
            copDealershipCollection[lastStake] = copDealershipCollection[tokenId];
            dealershipStakedCops.pop();
            delete copDealershipCollection[tokenId];
        }
    }

    //MISC: Emergency unstake tokens
    function rescue(uint256[] calldata tokenIds) external nonReentrant {
        require(rescueEnabled);
        uint256 tokenId;

        for (uint256 i = 0; i < tokenIds.length; i++) {
            tokenId = tokenIds[i];
            require(ownerOfStaked[tokenId] == msg.sender);
            if (citizen.isCop(tokenId)) {
                Stake memory lastStake = copStake[msg.sender][copStake[msg.sender].length - 1];
                copStake[msg.sender][copCollection[tokenId]] = lastStake;
                copCollection[lastStake.tokenId] = copCollection[tokenId];
                copStake[msg.sender].pop();

                totalCopsStaked--;
                delete copCollection[tokenId];
                delete ownerOfStaked[tokenId];
                citizen.safeTransferFrom(address(this), msg.sender, tokenId,"");
            } else {
                Stake memory lastStake = thugStake[msg.sender][thugStake[msg.sender].length - 1];
                thugStake[msg.sender][thugCollection[tokenId]] = lastStake;
                thugCollection[lastStake.tokenId] = thugCollection[tokenId];
                thugStake[msg.sender].pop();

                totalThugsStaked--;
                delete thugCollection[tokenId];
                delete ownerOfStaked[tokenId];
                citizen.safeTransferFrom(address(this),msg.sender,tokenId,"");
            }
        }
    }

    // Sets chance (if not using Cop % based model)
    function setChance(uint256 _chance) external {
        chance = _chance;
    }


    // Returns balance of users Thugs and Cops + deposited funds
    function getUserBalance(address _addr) external view virtual returns (uint256 balance) {
        balance = 0;
        uint256 currentBlock = block.timestamp;
        if(block.timestamp > lastBlockTimestamp){
            currentBlock = lastBlockTimestamp;
        }
        for(uint256 i = 0; i < thugStake[_addr].length; i++){
            balance += ((currentBlock - thugStake[_addr][i].value) * DAILY_BILLS_RATE) / 1 days - thugStake[_addr][i].spent;
        }
        for(uint256 i = 0; i < copStake[_addr].length; i++){
            balance += ((currentBlock - copStake[_addr][i].value) * DAILY_COP_BILLS_RATE) / 1 days - copStake[_addr][i].spent;
        }
        balance += deposited[_addr];
    }

    // Returns points of user
    function getUserPoints(address _addr) external view virtual returns (uint256 points) {
        points += getAllUnclaimedMedallions(_addr);
        points += assets.getPoints(_addr);
    }

    // Returns unclaimed medallions of specific tokenId
    function getUnclaimedMedallions(uint256 tokenId) external view virtual returns (uint256) {
        return locationPointTracker[copStake[ownerOfStaked[tokenId]][copCollection[tokenId]].location] - copStake[ownerOfStaked[tokenId]][copCollection[tokenId]].locationPoints;
    }

    // Returns all user unclaimed medallions
    function getAllUnclaimedMedallions(address addr) public view virtual returns (uint256 medallions) {
        for(uint i = 0; i < copStake[addr].length; i++){
            medallions += locationPointTracker[copStake[addr][i].location] - copStake[addr][i].locationPoints;
        }
    }

    // Returns array of Thug tokenIds staked by user
    function getUserStakedThugs(address _addr) external view virtual returns(uint256[] memory){
        uint256[] memory tokenIds = new uint256[](thugStake[_addr].length);
        for(uint256 i = 0; i < thugStake[_addr].length; i++){
            tokenIds[i] = thugStake[_addr][i].tokenId;
        }
        return tokenIds;
    }

    // Returns array of Cop tokenIds staked by user
    function getUserStakedCops(address _addr) external view virtual returns(uint256[] memory){
        uint256[] memory tokenIds = new uint256[](copStake[_addr].length);
        for(uint256 i = 0; i < copStake[_addr].length; i++){
            tokenIds[i] = copStake[_addr][i].tokenId;
        }
        return tokenIds;
    }

    // Returns array of users staked cops at location
    function getUserLocationStakedCops(address _addr, uint256 _location) public view virtual returns(uint256[] memory){
        uint256[] memory tokenIds = new uint256[](copStake[_addr].length);
        uint256 count = 0;
        for(uint256 i = 0; i < copStake[_addr].length; i++){
            if(copStake[_addr][i].location == _location){
                tokenIds[count] = copStake[_addr][i].tokenId;
                count++;
            }
        }
        return tokenIds;
    }

    //Returns all cops staked at location
    function getLocationStakedCops(uint256 _location) public view virtual returns (uint256[] memory tokenIds) {
        if(_location == 1){
            tokenIds = dealershipStakedCops;
        }else if(_location == 2){
            tokenIds = prisonStakedCops;
        }else if(_location == 3){
            tokenIds = casinoStakedCops;
        }else if(_location == 4){
            tokenIds = bankStakedCops;
        }
    }

    // Returns cop location
    function getCopLocation(uint256 tokenId) public view virtual returns(uint256){
        return copStake[ownerOfStaked[tokenId]][copCollection[tokenId]].location;
    }

    // SETTERS //

    //OWNER: Set prices of locations
    function setLocationPrices(uint256 _bank, uint256 _casino, uint256 _prison, uint256 _dealership, uint256 _gunCost) external onlyOwner {
        prices[3] = _bank;
        prices[2] = _casino;
        prices[1] = _prison;
        prices[0] = _dealership;
        gunCost = _gunCost;
    }

    //OWNER: Enabling rescue command
    function setRescueEnabled(bool _enabled) external onlyOwner {
        rescueEnabled = _enabled;
    }

    //OWNER: Sets timestamp of game end
    function setLastTimestamp(uint256 _timestamp) external onlyOwner {
        lastBlockTimestamp = _timestamp;
    }

    // Sets newest tokenid that can be staked (cops have been mapped up to this tokenId)
    function setNewestTokenId(uint256 _tokenId) external onlyOwner {
        newestTokenID = _tokenId;
    }

    //OWNER: Set contract addresses
    function setContracts(address _citizen, address _bills, address _assets) public onlyOwner {
        citizen = IThugCityNFT(_citizen);
        bills = IBills(_bills);
        assets = IAssets(_assets);
    }

    //OWNER: Enable pausing of game
    function setPaused(bool _state) external onlyOwner {
        _paused = _state;
    }

    //MISC: View if game is paused
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    modifier whenNotPaused() {
        require(!paused());
        _;
    }


    //MISC: Get random number
    function getRandomNumber(uint256 _seed, uint256 _limit) internal view returns (uint256) {
        uint256 extra = 0;
        for (uint256 i = 0; i < 7; i++) {
            extra += _randomSource[_randomIndex].balance;
        }

        uint256 random = uint256(
            keccak256(
                abi.encodePacked(
                    _seed,
                    blockhash(block.number - 1),
                    block.coinbase,
                    block.difficulty,
                    msg.sender,
                    extra,
                    _randomCalls,
                    _randomIndex
                )
            )
        );

        return uint256(random % _limit);
    }

    //OWNER: Change source of random numbers
    function changeRandomSource(uint256 _id, address _address) external onlyOwner {
        _randomSource[_id] = _address;
    }

    //MISC: Update random index
    function updateRandomIndex() internal {
        _randomIndex++;
        _randomCalls++;
        if (_randomIndex > 4) _randomIndex = 0;
    }

    //OWNER: Shuffle seeds of random numbers
    function shuffleSeeds(uint256 _seed, uint256 _max) external onlyOwner {
        uint256 shuffleCount = getRandomNumber(_seed, _max);
        _randomIndex = uint256(shuffleCount);
        for (uint256 i = 0; i < shuffleCount; i++) {
            updateRandomIndex();
        }
    }

    //MISC: ERC721 needed command
    function onERC721Received(address, address from, uint256, bytes calldata) external pure override returns (bytes4) {
        require(from == address(0x0));
        return IERC721Receiver.onERC721Received.selector;
    }
}

// File: Context.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: ERC165.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId)
        public
        view
        virtual
        override
        returns (bool)
    {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: ERC721.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./IERC721.sol";
import "./IERC721Receiver.sol";
import "./IERC721Metadata.sol";
import "./Address.sol";
import "./Context.sol";
import "./Strings.sol";
import "./ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId)
        public
        view
        virtual
        override(ERC165, IERC165)
        returns (bool)
    {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner)
        public
        view
        virtual
        override
        returns (uint256)
    {
        require(
            owner != address(0),
            "ERC721: balance query for the zero address"
        );
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId)
        public
        view
        virtual
        override
        returns (address)
    {
        address owner = _owners[tokenId];
        require(
            owner != address(0),
            "ERC721: owner query for nonexistent token"
        );
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId)
        public
        view
        virtual
        override
        returns (string memory)
    {
        require(
            _exists(tokenId),
            "ERC721Metadata: URI query for nonexistent token"
        );

        string memory baseURI = _baseURI();
        return
            bytes(baseURI).length > 0
                ? string(abi.encodePacked(baseURI, tokenId.toString()))
                : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId)
        public
        view
        virtual
        override
        returns (address)
    {
        require(
            _exists(tokenId),
            "ERC721: approved query for nonexistent token"
        );

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved)
        public
        virtual
        override
    {
        require(operator != _msgSender(), "ERC721: approve to caller");

        _operatorApprovals[_msgSender()][operator] = approved;
        emit ApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator)
        public
        view
        virtual
        override
        returns (bool)
    {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(
            _isApprovedOrOwner(_msgSender(), tokenId),
            "ERC721: transfer caller is not owner nor approved"
        );

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        require(
            _isApprovedOrOwner(_msgSender(), tokenId),
            "ERC721: transfer caller is not owner nor approved"
        );
        _safeTransfer(from, to, tokenId, _data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `_data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(
            _checkOnERC721Received(from, to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId)
        internal
        view
        virtual
        returns (bool)
    {
        require(
            _exists(tokenId),
            "ERC721: operator query for nonexistent token"
        );
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner ||
            getApproved(tokenId) == spender ||
            isApprovedForAll(owner, spender));
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(
            ERC721.ownerOf(tokenId) == from,
            "ERC721: transfer of token that is not own"
        );
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        if (to.isContract()) {
            try
                IERC721Receiver(to).onERC721Received(
                    _msgSender(),
                    from,
                    tokenId,
                    _data
                )
            returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert(
                        "ERC721: transfer to non ERC721Receiver implementer"
                    );
                } else {
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}
}

// File: IERC165.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: IERC721.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(
        address indexed from,
        address indexed to,
        uint256 indexed tokenId
    );

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(
        address indexed owner,
        address indexed approved,
        uint256 indexed tokenId
    );

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(
        address indexed owner,
        address indexed operator,
        bool approved
    );

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId)
        external
        view
        returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator)
        external
        view
        returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}

// File: IERC721Metadata.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: IERC721Receiver.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "./Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(
        address indexed previousOwner,
        address indexed newOwner
    );

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _setOwner(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _setOwner(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(
            newOwner != address(0),
            "Ownable: new owner is the zero address"
        );
        _setOwner(newOwner);
    }

    function _setOwner(address newOwner) private {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}
// File: Strings.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length)
        internal
        pure
        returns (string memory)
    {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}